

## F07TUF (CTRCON/ZTRCON) – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

F07TUF (CTRCON/ZTRCON) estimates the condition number of a complex triangular matrix.

### 2 Specification

```

SUBROUTINE F07TUF(NORM, UPLO, DIAG, N, A, LDA, RCOND, WORK, RWORK,
1              INFO)
ENTRY      ctrcon(NORM, UPLO, DIAG, N, A, LDA, RCOND, WORK, RWORK,
1              INFO)
INTEGER    N, LDA, INFO
real      RCOND, RWORK(*)
complex  A(LDA,*), WORK(*)
CHARACTER*1 NORM, UPLO, DIAG

```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine estimates the condition number of a complex triangular matrix  $A$ , in either the 1-norm or the infinity-norm:

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1 \quad \text{or} \quad \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty.$$

Note that  $\kappa_\infty(A) = \kappa_1(A^T)$ .

Because the condition number is infinite if  $A$  is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine computes  $\|A\|_1$  or  $\|A\|_\infty$  exactly, and uses Higham's implementation of Hager's method [1] to estimate  $\|A^{-1}\|_1$  or  $\|A^{-1}\|_\infty$ .

### 4 References

- [1] Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

### 5 Parameters

- 1: NORM — CHARACTER\*1 *Input*  
*On entry:* indicates whether  $\kappa_1(A)$  or  $\kappa_\infty(A)$  is estimated as follows:  
     if NORM = '1' or 'O', then  $\kappa_1(A)$  is estimated;  
     if NORM = 'I', then  $\kappa_\infty(A)$  is estimated.  
*Constraint:* NORM = '1', 'O' or 'I'.
- 2: UPLO — CHARACTER\*1 *Input*  
*On entry:* indicates whether  $A$  is upper or lower triangular as follows:  
     if UPLO = 'U', then  $A$  is upper triangular;  
     if UPLO = 'L', then  $A$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.

- 3:** DIAG — CHARACTER\*1 *Input*  
*On entry:* indicates whether  $A$  is a non-unit or unit triangular matrix as follows:  
 if DIAG = 'N', then  $A$  is a non-unit triangular matrix;  
 if DIAG = 'U', then  $A$  is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.  
*Constraint:* DIAG = 'N' or 'U'.
- 4:** N — INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 5:** A(LDA,\*) — *complex* array *Input*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1,N)$ .  
*On entry:* the  $n$  by  $n$  triangular matrix  $A$ . If UPLO = 'U',  $A$  is upper triangular and the elements of the array below the diagonal are not referenced; if UPLO = 'L',  $A$  is lower triangular and the elements of the array above the diagonal are not referenced. If DIAG = 'U', the diagonal elements of  $A$  are not referenced, but are assumed to be 1.
- 6:** LDA — INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F07TUF (CTRCON/ZTRCON) is called.  
*Constraint:*  $LDA \geq \max(1,N)$ .
- 7:** RCOND — *real* *Output*  
*On exit:* an estimate of the reciprocal of the condition number of  $A$ . RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than *machine precision*, then  $A$  is singular to working precision.
- 8:** WORK(\*) — *complex* array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1,2*N)$ .
- 9:** RWORK(\*) — *real* array *Workspace*  
**Note:** the dimension of the array RWORK must be at least  $\max(1,N)$ .
- 10:** INFO — INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed estimate RCOND is never less than the true value  $\rho$ , and in practice is nearly always less than  $10\rho$ , although examples can be constructed where RCOND is much larger.

## 8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form  $Ax = b$  or  $A^H x = b$ ; the number is usually 5 and never more than 11. Each solution involves approximately  $4n^2$  real floating-point operations but takes considerably longer than a call to F07TSF (CTRTRS/ZTRTRS) with 1 right-hand side, because extra care is taken to avoid overflow when  $A$  is approximately singular.

The real analogue of this routine is F07TGF (STRCON/DTRCON).

## 9 Example

To estimate the condition number in the 1-norm of the matrix  $A$ , where

$$A = \begin{pmatrix} 4.78 + 4.56i & 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.00 - 0.30i & -4.11 + 1.25i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.89 - 1.34i & 2.36 - 4.25i & 4.15 + 0.80i & 0.00 + 0.00i \\ -1.89 + 1.15i & 0.04 - 3.69i & -0.02 + 0.46i & 0.33 - 0.26i \end{pmatrix}.$$

The true condition number in the 1-norm is 70.27.

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07TUF Example Program Text
*      Mark 16 Release. NAG Copyright 1993.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDA
      PARAMETER        (NMAX=8,LDA=NMAX)
      CHARACTER        NORM, DIAG
      PARAMETER        (NORM='1',DIAG='N')
*      .. Local Scalars ..
      real             RCOND
      INTEGER          I, INFO, J, N
      CHARACTER        UPLO
*      .. Local Arrays ..
      complex         A(LDA,NMAX), WORK(2*NMAX)
      real            RWORK(NMAX)
*      .. External Functions ..
      real            X02AJF
      EXTERNAL         X02AJF
*      .. External Subroutines ..
      EXTERNAL         ctrcon
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07TUF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*

```

```

*       Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF

*
*       Estimate condition number
*
      CALL ctrcon(NORM,UPLO,DIAG,N,A,LDA,RCOND,WORK,RWORK,INFO)

*
      WRITE (NOUT,*)
      IF (RCOND.GE.X02AJF()) THEN
        WRITE (NOUT,99999) 'Estimate of condition number =',
+       1.0e0/RCOND
      ELSE
        WRITE (NOUT,*) 'A is singular to working precision'
      END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,e10.2)
      END

```

## 9.2 Program Data

F07TUF Example Program Data

```

4                                     :Value of N
'L'                                   :Value of UPLO
( 4.78, 4.56)
( 2.00,-0.30) (-4.11, 1.25)
( 2.89,-1.34) ( 2.36,-4.25) ( 4.15, 0.80)
(-1.89, 1.15) ( 0.04,-3.69) (-0.02, 0.46) ( 0.33,-0.26) :End of matrix A

```

## 9.3 Program Results

F07TUF Example Program Results

Estimate of condition number = 3.74E+01